Customforce Business Processes
100 Sample Formulas to Get You Started

100 packaged formulas for use in salesforce.com applications, including:

- Case Management
- Commission Calculations
- Contact Management
- Data Categorization
- Data Validation
- Date Calculations
- Discounting
- Expense Tracking
- Financial Calculations
- Integration Links
- Lead Management
- Metrics
- News Services
- Opportunity Management
- Pricing
- Project Management
- Scoring Calculations
- Tax Calculations
- Time Tracking
- Web Site Links

For in-depth information, go to the Custom Formulas home page on www.crmsuccess.com.

sf_fqrg_ind_v6

Case Management

Round-Robin Case Assignment
Evenly distributes cases to a pool of users using a simple round-robin algorithm. This scenario applies the MOD function to the Case Number field, to produce a number that repeatedly cycles between 1 and the number of users in the round robin pool (3 in this example). After creating this formula field, you must create case assignment rules based on it. For each value of the formula field (1, 2, or 3 in this example), you specify the user or queue assignment.

MOD(VALUE({!CaseNumber}), 3) + 1

Case Aging (for open cases)
Calculates the number of days a case has been open. If the case is closed, sets the result to blank. Add this formula to a related list as the sort column, and you can quickly see which currently open cases have been open the longest.

IF({!IsClosed},null, NOW() - {!CreatedDate})

Case Categorization
Displays a text value of RED, YELLOW, or GREEN, depending on case age (a custom field). This formula could be used for grouping in a summary report.

IF({!Case_Age__c} > 20, "RED", IF({!Case_Age__c} > 10, "YELLOW", "GREEN"))

Case Due Date Calculation
Sets the due date based on the priority of the case. If it is high, due date is 2 days after being opened. If it is medium, due date is 5 days after opening. Otherwise, due date is 7 days.

IF(ISPICKVAL({!Priority}, "High"), {!CreatedDate} + 2, IF(ISPICKVAL({!Priority}, "Medium"), {!CreatedDate} + 5, {!CreatedDate} + 7))

Autodial
Creates a linkable phone number field that automatically dials the phone number when clicked. In this example, replace "servername" and "call" with the name of your dialing tool and the command it uses to dial. The merge field, {!Id} inserts the identifier for the contact, lead, or account record. The first {!Phone} merge field tells the dialing tool the number to call, and the last {!Phone} merge field uses the value of the Phone field as the linkable text the user clicks to dial.

HYPERLINK("http://servername/call?id=" & {!Id} & "&phone=" & {!ContactPhone}, {!ContactPhone})

Case Aging (all cases)
Calculates the number of days a case has been open, for both open and closed cases.

IF({!IsClosed} , ROUND({!ClosedDate} - {!CreatedDate}, 0) , ROUND((NOW() - {!CreatedDate}),0))

Case Management (continued)

Suggested Offers
Suggests a product based on support history for a computer reseller. When problem (custom field) matches a field, a suggestion can be offered.

```
CASE( {!Problem__c} , "Memory", "Suggest new memory cards", "Hard Drive failure", "Suggest new hard drive with tape backup", "")
```

Suggested Agent Prompts
Prompts an agent with cross-sell offers based on past purchases.

```
CASE( {!Product_Purch__c} , "Printer", "Extra toner cartridges", "Camera", "Memory cards", "Special of the day")
```

Commission Calculations

Flat Rate Commission Calculation
Calculates a commission based on an Opportunity amount and percentage. This is a simple scenario where commission is a flat 8% of Opportunity amount for Closed Won Opportunities.

```
IF(ISPICKVAL( {!StageName}, "Closed Won"), ROUND({!Amount} * 0.08, 2), 0)
```

Rule-Based Commission Calculations
Calculates a commission rate based on deal size. 9% commission paid for Large Deals.

```
IF( {!Amount} > 100000, 0.09, 0.08 )
```

Contact Management

Contact Age (years old)
Calculates current age of contact in years. Subtracts birth date from today’s date, resulting in a number in decimal days. Divides this number by 365 to get years. Uses floor function to drop the decimal portion.

```
FLOOR((TODAY() - {!Birthdate})/365)
```

Birthday This Month Indicator
Displays the value "Yes" if the contact’s birthday falls in the current calendar month.

```
IF ( MONTH({!Birthdate}) = MONTH(TODAY()), "Yes", "")
```

Preferred Contact Phone Number
In a contact-related list, displays the contact’s preferred contact method -- work phone, home phone, or mobile phone– based on selected option in a Preferred Phone pick list.

```
IF( ISPICKVAL( {!Preferred_Phone__c} , "Work"), "w. " & {!Phone} ), IF( ISPICKVAL( {!Preferred_Phone__c} , "Home"), "h. " & {!HomePhone} ), IF( ISPICKVAL( {!Preferred_Phone__c} , "Mobile"), "m. " & {!MobilePhone} ), "No Preferred Phone")))
```

Contact Management (continued)

Contract Aging
Calculates the number of days since the contract was activated. If the contract status is not "Activated," this field is blank.

IF(ISPICKVAL({!Status} , "Activated"), NOW()-{!ActivatedDate}, null)

Contact Identification Numbering
Displays the first 5 characters of the contact's last name and the last 4 characters of the contact's social security number separated by a dash. Note that this example uses a text custom field called SSN on contacts.

TRIM(LEFT({!LastName}, 5))&"-"&TRIM(RIGHT({!SSN__c}, 4))

Expiration Month
Returns the month that your service-level agreement expires. This example uses a custom date field called SLA Expiration Date.

MONTH({!SLAExpirationDate__c})

Telephone Country Code
Determines the telephone country code based on the country of the mailing address. Could be modified to use Billing country or other country field.

CASE({!MailingCountry} , "USA", "1", "Canada", "1", "France", "33", "UK", "44", "Australia", "61", "Japan", "81", "?")

Dynamic Address Formatting
Displays City, State, Zip, or Postal Code City in standard format -- depending on country.

CASE({!ShippingCountry} , "USA", {!ShippingCity} & ", " & {!ShippingState} & " " & {!ShippingPostalCode} , "France", {!ShippingPostalCode} & " " & {!ShippingCity} , "etc")

Contact Autodial
Creates a dynamic hyperlink that passes the contact record ID and phone number to the Sforce TAPI toolkit for phone dialer integration.

HYPERLINK("http://localhost:8541/call?id=" & {!Id} & "&phone=" & {!Phone} & "&sid=" & GETSESSIONID(), {!Phone})

Unformatted Phone Number
Removes the parentheses and dash format characters from North American phone numbers. This is necessary for some auto dialer software.

IF({!Country_Code__c} = "1" , MID({!Phone} ,2, 3) & MID({!Phone},7,3) & MID({!Phone},11,4), {!Phone})

Data Categorization

Aging of Contract Approval Process
Calculates how many days a contract is in the approval process. This example is a number formula field on contracts that uses a custom date field called Date in Approval.

TODAY()-{!Date_in_approval__c}

Data Categorization (continued)

Abbreviation Codes
Displays the text "Medical" for any Product Code that begins with "ICU." For all other products, displays "Technical."

IF(BEGINS({!ProductCode} , "ICU"), "Medical", "Technical")

Revenue Categorization
Assigns the value "1" to accounts with revenues of less than one million dollars and the value "2" to accounts with revenues greater than one million.

IF({!AnnualRevenue} < 1000000, 1, 2)

Deal Size Categorization (small deal examples)
Displays "Small" if the price and quantity are less than one. This field is blank if the asset has a price or quantity greater than one.

IF(AND({!Price}<1,{!Quantity}<1),"Small", null)

Product Categorization
Displays "Parts" for any product with the word "part" in it. Otherwise, displays "Service."

IF(CONTAINS({!ProductCode} , "part"), "Parts", "Service")

Deal Size Categorization (large deals)
Displays "Large" for deals over one million dollars.

IF({!Price__c} > 1000000, "Large Deal", "Small Deal")

Data Validation

Case Data Completeness Tracking
Calculates the percent of your important case fields that are being filled by your support personnel. This formula field checks 2 fields to see if they are blank. If so, a zero is counted for that field. A "1" is counted for any field that contains a value and this total is divided by 2 (the number of fields evaluated). The fields Problem_Num and Severity_Num must be numeric.

(IF(ISNULL({!Problem_Num__c}), 0, 1) + IF(ISNULL({!Severity_Num__c}), 0,1)) / 2

Invalid Renewal Expiration Date
Checks if Renewal Expiration Date is after Close Date (custom field). If true, it displays an error message; otherwise it is blank. This formula can be used in a custom report that lists error records, or it can be used in a workflow rule to assign a task to the owner to correct the error.

IF ({!Expiration_Date__c} > {!CloseDate}, "Error: renewal expiration date cannot be after Close Date", "")

Data Validation (continued)

Invalid Discount
Checks if discount percent field (custom field) is between 0 and 50%. If not, it displays an error message; otherwise it is blank. This formula can be used in a custom report that lists error records, or it can be used in a workflow rule to assign a task to the owner to correct the error.

IF ({!Discount_Percent__c} > 0.50, "Error: Discount cannot exceed 50%",
IF ({!Discount_Percent__c} < 0, "Error: Discount cannot be less than 0", ""))

Gratitude Checker
Checks if message field (custom field) contains the words "Thanks" or "Thank you". If not, it displays an error message; otherwise it is blank. This formula can be used in a custom report that lists error records, or it can be used in a workflow rule to assign a task to the owner to correct the rudeness.

IF(OR(CONTAINS({!Message__c}, "Thanks"), CONTAINS({!Message__c}, "Thank you")), "", "Did not express gratitude in message.")

Date Calculations

Day of the Year
Calculates today's numeric day of the year (a number between 1 and 365)

TODAY() – DATE(YEAR(TODAY()), 1, 1) + 1

Day of Week (number)
Calculates today's day of the week as a number (0 = Sunday, 1 = Monday, 2 = Tuesday, etc.). This formula can be applied to any date field by substituting the date field instead of the TODAY function.

MOD(TODAY() - DATE(1900, 1, 7), 7)

Day of Week (text)
Calculates today's day of the week and displays as text. This formula can be applied to any date field by substituting the date field instead of the TODAY function.

CASE(MOD(TODAY() - DATE(1900, 1, 7), 7), 0, "Sunday", 1, "Monday", 2, "Tuesday", 3, "Wednesday", 4, "Thursday", 5, "Friday", 6, "Saturday","Error")

Last Activity Month Display
Displays the month of the last account activity or is blank if there are no activities for the account.

CASE(MONTH({!LastActivityDate}) ,1, "January", 2, "February", 3, "March", 4, "April", 5, "May", 6, "June", 7, "July", 8, "August", 9, "September", 10, "October", 11, "November", 12, "December", "blank")

Days Until End of Month
Displays the number of days between a specific date and the end of the month that date occurs in.

IF(MONTH({!CloseDate})=12, DATE(YEAR({!CloseDate}),12,31)-{!CloseDate}, DATE(YEAR({!CloseDate}),MONTH({!CloseDate})+1,1)-{!CloseDate}-1)

Discounting

Opportunity Discounting
Calculates a discounted Opportunity amount using a simple percentage. This is a simplistic scenario where discount is applied at the Opportunity level, rather than the Opportunity Line level. Discount Percent is a custom field on Opportunity.

ROUND({!Amount} – {!Amount} * {!Discount-Percent__c}, 2)

Expense Tracking

Mileage Expense Calculation
Calculates mileage expenses associated with visiting a customer site, at 35 cents per mile.

Mileage Expense = {!Miles_Driven__c} * 0.35

Expense Number String Connect
Displays the text "Expense# " followed by the expense number. This is a text formula field that uses an expense number custom field.

"Expense# "& {!ExpenseNumber__c}

Financial Calculations

Compound Interest
Calculates interest, compounded m times per year, you will have after t years.

{!P} * (1 + {!R} / {!M}) ^ ({!T} * {!M})

Continuously Compounding Interest
Calculates interest, if continuously compounded, you will have after t years.

{!P} * EXP({!R} * {!T})

Payment Status
Determines if the payment due date is past and the payment status is "UNPAID." If so, returns the text "PAYMENT OVERDUE" and if not, leaves the field blank. This example uses a custom date field called Payment Due Date and a text custom field called Payment Status on contracts.

IF(AND({!Payment_Due_Date__c} < TODAY(),{!Payment_Status__c} ="UNPAID") , "PAYMENT OVERDUE", null)

Payment Due Indicator
Returns the date five days after the contract start date whenever Payment Due Date is blank. Payment Due Date is a custom date field on contracts.

NULLVALUE({!Payment_Due_Date__c} , {!Start-Date} +5)

Financial Calculations (continued)

Gross Margin (based on percent)
Calculates gross margin.

{!Margin_percent__c} * {!Items_Sold__c} * {!Price_item__c}

Gross Margin (with COGS)
Provides a simple calculation of gross margin when cost of goods sold is known as custom field.

{!Amount} - {!Cost__c}

Rules-based Status Message
Uses conditional logic to display a "Payment Overdue" status message for open invoices.

IF(AND({!PaymentDueDate__c} < TODAY(), ISPICKVAL({!PaymentStatus},"UNPAID")), "PAYMENT OVERDUE", "")

Integration Links

Application API Link
Creates a link to an application outside Salesforce, passing the parameters so that it can connect to Salesforce via the API and create the necessary event.

HYPERLINK("https://www.myintegration.com?sId="& GETSESSIONID() & "?&rowID="&{!Name} & "action=CreateTask", "Create a Meeting Request")

Skype Auto Dialer Integration
Creates a linkable phone number field that automatically dials the phone number via the Skype VOIP phone application. This example requires installation of the Skype application (a third-party product not provided by salesforce.com) on your desktop and modification of your PC's registry to associate the "callto:" protocol with the Skype application.

HYPERLINK("callto://+" & {!Country_Code__c} & {!Phone_Unformatted__c}, {!Phone})

Lead Management

Autodial
Creates a linkable phone number field that automatically dials the phone number when clicked. In this example, replace "servername" and "call" with the name of your dialing tool and the command it uses to dial. The merge field, {!Id} inserts the identifier for the contact, lead, or account record. The first {!Phone} merge field tells the dialing tool the number to call, and the last {!Phone} merge field uses the value of the Phone field as linkable text the user clicks to dial.

HYPERLINK("http://servername/call?id=" & {!Id} & "&phone=" & {!Phone}, {!Phone})

Lead Management (continued)

Round-Robin Lead Assignment
Evenly distributes leads to a pool of users using a simple round-robin algorithm. This scenario applies the MOD function to the Lead Number field (a custom autonumber field with display format of {000000}, to produce a number that repeatedly cycles between 1 and the number of users in the round robin pool (3 in this example). After creating this formula field, you must create lead assignment rules based on it. For each value of the formula field (1, 2, or 3 in this example), you specify the user or queue assignment.

```
MOD( VALUE({!Lead_Number__c} ), 3) + 1
```

Lead Aging (for open leads)
Checks to see if a lead is open and, if so, calculates the number of days it has been open by subtracting the date and time created from the current date and time. The result is the number of days open rounded to 0 decimal places. If the lead is not open, this field is blank.

```
IF(ISPICKVAL({!Status} , "Open"), ROUND(NOW()-{!CreatedDate}, 0), NULL)
```

Lead Data Completeness
Calculates the percent of your important lead fields that are being filled by your sales personnel. This formula field checks 2 fields to see if they are blank. If so, a zero is counted for that field. A "1" is counted for any field that contains a value and this total is divided by 2 (the number of fields evaluated). Lead_Num must be a numeric.

```
( IF({!Phone} = "", 0, 1) + IF({!Email} = "", 0, 1) ) / 2
```

Lead Numbering
Returns a number for the text value in the auto-number field "Lead Number." This can be useful if you want to use the "Lead Number" field in a calculation, such as for round-robin or other routing purposes. Note that auto-number fields are text fields and must be converted to a number for numeric calculations.

```
VALUE( {!Lead_Number__c} )
```

Metrics

Unit of Measure Conversion
Converts kilometers to miles.

```
{!Miles__c}/.6
```

News Services

Google Search Creates a Google search on Account Name.	HYPERLINK("http://www.google.com/search?en&q="&{!Account_Name}, "Google")
Google News Search Company Name Creates a Google news search on Account Name.	HYPERLINK("http://www.google.com/news?en&q="&{!Account_Name}, "Google News")
Yahoo News Search Creates a Yahoo news search on Account Name.	HYPERLINK("http://news.search.yahoo.com/search/news?p="&{!Account_Name}, "Yahoo News")
Yahoo Search Creates a Yahoo search on Account Name.	HYPERLINK("http://search.yahoo.com/search?p="&{!Account_Name}, "Yahoo Search")
MarketWatch Search Searches for ticker symbol and information on Marketwatch.com using custom field TICKER_SYMBOL.	HYPERLINK("http://www.marketwatch.com/tools/quotes/quotes.asp?symb="&{!TickerSymbol},"Marketwatch")
BBC News Searches for company information on BBC news on Account_name.	HYPERLINK("http://newssearch.bbc.co.uk/cgi-bin/search/results.pl?scope=newsifs&tab=news&q="&{!Account_name}, "BBC News")
CNN News Search Searches for company information on the CNN Web site on Account_name.	HYPERLINK("http://websearch.cnn.com/search/search?source=cnn&invocationType=search%2Ftop&sites=web&query="&{!Account_name}, "CNN News")
Forbes News Search Searches for company information on Forbes online on Account_name.	HYPERLINK("http://www.forbes.com/search/results.jhtml?MT="&{!Account_name},"Forbes News")
Bloomberg News Search Searches for Ticker Symbol of a company on the Bloomberg Web site.	HYPERLINK("http://www.bloomberg.com/apps/quote?ticker="&{!TickerSymbol}, "Bloomberg News")
MSNBC News Search Searches for company information on MSNBC by Account_name.	HYPERLINK("http://search.msn.com/results.asp?cp=1252&q="&{!Account_name},"MSN News")

Opportunity Management

Create Event Adds a link called "Create Event" that, when clicked, creates a new event that is associated with the current opportunity.	HYPERLINK("00U/e?retURL=%2F006x0000001T8Om&what_id=" & {!Id}, "Create Event")
Opportunity Data Completeness Calculates the percentage of your important opportunity fields that are being filled by your sales personnel. This formula field checks 5 fields to see if they are blank. If so, a zero is counted for that field. A "1" is counted for any field that contains a value and this total is divided by 5 (the number of fields evaluated).	(IF(ISNULL({!Maint_Amount__c}), 0, 1) + IF(ISNULL({!Services_Amount__c}), 0,1) + IF(ISNULL({!Discount_Percent__c}), 0, 1) + IF(ISNULL({!Amount}), 0, 1)+IF(ISNULL({!Timeline__c}), 0, 1)) / 5
Days Left to Close Returns the number of days left to close.	{!Expected_close_date__c} -TODAY()
Opportunity Discounting with Approval Add a "Discount Approved" checkbox to Opportunity. Sets up FLS so that only a sales ops exec can approve discounts. Uses conditional logic to check the value of the approval flag before calculating the commission.	IF({!Discount_Approved__c, ROUND({!Amount} – {!Amount} * {!Discount-Percent__c}, 2), {!Amount})
Maintenance Calculation Calculates maintenance fees as 20% of license fees per year. Maintenance Years is a custom field on Opportunity.	{!Amount} * {!Maint_Years__c} * 0.2
Total Deal Size Calculates the sum of maintenance and services amounts.	{!Amount} + {!Maint_Amount__c} + {!Services_Amount__c}
Monthly Subscription-based Calculated Amounts Calculate an opportunity amount based on a monthly subscription rate multiplied by the subscription period.	{!Monthly_Amount__c} * {!Subscription_Months__c}
Rules-based Opportunity Categorization Uses conditional logic to populate an Opportunity category text field, based on amount. Opportunities with amounts less than $1500 are "Category 1," between $1500 and $10000 are "Category 2," and the rest are "Category 3." This example uses nested IF statements.	IF({!Amount} < 1500, "Category 1", IF({!Amount} > 10000, "Category 3", "Category 2"))
Opportunity Split Credit for Sales Reps Splits opportunity amount between multiple reps. "Total reps" indicates the total number of reps on the deal.	{!Amount} / {!total_reps__c}

Opportunity Management (continued)

Completed Opportunity Fields
Calculates the percentage of your important opportunity fields that are being filled by your sales personnel. This formula field checks 5 fields to see if they are blank. If so, a zero is counted for that field. A "1" is counted for any field that contains a value, and this total is divided by 5 (the number of fields evaluated).

(IF(ISNULL({!Maint_Amount__c}), 0, 1) + IF(ISNULL({!Services_Amount__c}), 0,1) + IF(ISNULL({!Discount_Percent__c}), 0, 1) + IF(ISNULL({!Amount}), 0, 1)+IF(ISNULL({!Timeline__c}), 0, 1)) / 5

Revenue Display (no "$")
Returns the expected revenue amount of an opportunity in text format without a dollar sign. For example, if the Expected Revenue of a campaign is "$200,000," this formula field displays "200000."

TEXT({!ExpectedRevenue})

Total Price (based on units)
Generates proposal pricing based on unit price and total volume.

{!Unit_price__c} * {!Volume__c} * $20

Expected License Revenue
Calculates expected revenue for licenses based on probability of closing.

{!Expected_rev_licenses__c}*{!Prob__c}

User Pricing
Calculates price per user license.

{!Total_license_rev__c} / {!Number_user_licenses__c}

Margin Percent
Calculates total margin on an opportunity.

{!Sales_Price__c} / {!List_Price__c}

Expected Product Revenue
Calculates total revenue from multiple products, each with a different probability of closing.

{!ProductA_probability__c} * {!ProductA_revenue__c} + {!ProductB_probability__} * {!ProdB_revenue__c}

Total Contract Value from Recurring and Non-Recurring Revenue
Calculates both recurring and non-recurring revenue streams over lifetime of contract.

{!Non_Recurring_Revenue__c} + {!Contract_Length__c}*{!Recurring_Revenue__c}

Shipping Cost Percentage
Calculates shipping cost as a fraction of total amount.

{!Ship_cost__c}/{!total_amount__c}

Monthly value
Divides total yearly value by 12 months.

{!Total_value__c} / 12

Opportunity Management (continued)

Opportunity Reminder Date
Creates reminder date based on 7 days before close date. Can be linked to workflow to trigger action.

Reminder Date = {!CloseDate} - 7

Display Close Month for Reporting Purposes
Creates a text formula with the Close Month spelled out, and then builds a custom report that groups by Close Month.

CASE(MONTH({!CloseDate}) , 1, "January", 2, "February", 3, "March", 4, "April", 5, "May", 6, "June", 7, "July", 8, "August", 9, "September", 10, "October", 11, "November", 12, "December", "Invalid month")

Shipping Cost by Weight
Calculates postal charges based on weight.

{!package_weight__c}*{cost_lb__c}

Account Rating
Example of an Account Rating formula. Evaluates annual revenue, billing country, and account type, and assigns a value of hot or cold.

IF(AND({!AnnualRevenue} > 10000000, CONTAINS(CASE({!BillingCountry}, "United States", "US", "America", "US", "USA", "US", "NA"), "US")), IF(ISPICKVAL({!Type} , "Manufacturing Partner"), "Hot", IF(OR(ISPICKVAL({!Type} , "Channel Partner / Reseller"),ISPICKVAL({!Type} , "Installation Partner")), "Warm", "Cold")), "Cold")

Professional Services Calculation
Estimates professional service fees at an average loaded rate of $1200 per day. Consulting Days is a custom field on Opportunity.

{!Consulting_Days__c} * 1200

Tiered Commission Rates
Calculates the 2% commission amount of an opportunity that has a probability of 100%. All other opportunities will have a commission value of 0.

IF({!Probability} =1, ROUND({!Amount}*0.02, 2) , 0)

Pricing

Total Amount
Calculates total amount based on unit pricing and total units.

{!Unit_price} * {!Price_unit}

Discounting
Displays "DISCOUNTED" on an opportunity if its maintenance amount and services amount do not equal the opportunity amount. Otherwise, displays "FULL PRICE." Note that this example uses two custom currency fields: Maint Amount and Services Amount.

IF({!Maint_Amount__c} + {!Services_Amount__c} <> {!Amount}, "DISCOUNTED", "FULL PRICE")

Project Management

Calculate Intermediate Milestone from End Date
Calculates intermediate milestone dates by subtracting days from end date (for projects that are planned based on end date).

Begin Test Phase = {!Release_Date__c} - 7 * 8

Scoring Calculations

Lead Scoring
Scores leads, providing a higher score for phone calls than Web requests.

Case({!Source}, "Phone", 2, "Web", 1, 0)

Customer Success Scoring
Uses a simple scoring algorithm to score customers high for positive survey results that have been entered into Salesforce.

{!Survey_Question_1__c} *5 + {!Survey_Q2__c} *2

Customer Satisfaction Scoring
Uses a simple scoring algorithm to score customer satisfaction based on survey results entered.

{!Survey_Cust_Sat} * 100

Tax Calculations

Federal Tax
Calculates expected federal tax.

{!Amount} * {!fed_rate__c}

State Tax
Calculates expected state tax.

{!Amount}*{!state_rate__c}

Local Tax
Calculates expected local tax.

{!Amount}*{!local_rate__c}

Time Tracking

Timesheet (hours worked per week)
Uses a custom tab to enable time tracking of hours worked per day. Uses a formula field to add up the hours per week.

{!MonHours__c} + {!TuesHours__c} + {!WedsHours__c} + {!ThursHours__c} + {!FriHours__c}

Time Tracking (continued)

Calculates Overtime Hour Pay
Determines total pay by calculating regular hours x regular pay rate, plus overtime hours x overtime pay rate.

Total Pay = IF({!Total_Hours__c} <= 40, {!Total_Hours__c} * {!Hourly_Rate__c}, 40 * {!Hourly_Rate__c} + ({!Total_Hours__c} - 40) * {!Overtime_Rate__c})

Web Site Links

Web Site Link
Creates a link named "Company's Website" that connects a user to the Web site address on the account record.

HYPERLINK({!Website}, "Company's Website")